
# Copyright (c) 2011 Nokia Corporation.

QT       += core gui declarative network
CONFIG   += qt-components

TARGET = socketdemo
TEMPLATE = app

VERSION = 0.0.1

SOURCES += main.cpp \
    serversocket.cpp \
    clientsocket.cpp

HEADERS  += \
    serversocket.h \
    clientsocket.h

# Publish the app version to source code.
DEFINES += APP_VERSION=\\\"$$VERSION\\\"

# Symbian specific
symbian {
    message(Symbian build)
    TARGET = SocketDemo
    TARGET.UID3 = 0xea6c2794
    TARGET.CAPABILITY += NetworkServices
    TARGET.EPOCSTACKSIZE = 0x14000
    TARGET.EPOCHEAPSIZE = 0x1000 0x1800000 # 24MB
    #ICON = icon.svg

    # Define the preprocessor macro to get the version in our app.
    contains(SYMBIAN_VERSION, Symbian3) {
        message(Symbian^3 build)
        DEFINES += APP_VERSION=\"$$VERSION\"
    }
    else {
        message(Symbian^1 build)
        DEFINES += Q_OS_SYMBIAN_1
    }

    OTHER_FILES += qml/SocketDemo/*.qml

    qmlfiles.sources = qml
    DEPLOYMENT += qmlfiles
}


# Simulator
simulator {
    message(Simulator build)

    OTHER_FILES += qml/SocketDemo/*.qml

    system(mkdir ..\\SocketDemo-build-simulator\\qml\\symbian)
    system(mkdir ..\\SocketDemo-build-simulator\\qml\\common)
    system(copy qml\\SocketDemo\\*.* ..\\SocketDemo-build-simulator\\qml\\SocketDemo)
}





